
FUNCTION TSE_User

LPARAMETERS cLogiciel, cVarEnv

LOCAL cRacine, cCheminUtilisateur

M.cRacine = ADDBS(SET("DIRECTORY"))

IF NOT EMPTY(M.cVarEnv)
   M.cCheminUtilisateur = GETENV(M.cVarEnv)
ENDIF

IF EMPTY(M.cCheminUtilisateur)
   M.cCheminUtilisateur = GETENV('HOMEDRIVE') + GETENV('HOMEPATH')
ENDIF

M.cCheminUtilisateur = ADDBS(M.cCheminUtilisateur)

IF NOT INLIST (MKDIR(M.cCheminUtilisateur + M.cLogiciel),0,17)
   Alerte("Impossible de créer le répertoire " + ;
          M.cCheminUtilisateur + M.cLogiciel + " pour l'utilisateur !")
   CANCEL
ENDIF
            
WAIT WINDOW "Vérification des fichiers utilisateurs..." NOWAIT

M.cErr = ON("ERROR")
ON ERROR DO Type_Erreur WITH ERROR()
   
DO CopyOnlyNew WITH '', M.cCheminUtilisateur + M.cLogiciel, "*.APP"
DO CopyOnlyNew WITH '', M.cCheminUtilisateur + M.cLogiciel, "*.DLL"
DO CopyOnlyNew WITH '', M.cCheminUtilisateur + M.cLogiciel, "*.FLL"

IF NOT EMPTY(M.cErr)
   ON ERROR &cErr
ELSE
   ON ERROR
ENDIF

WAIT CLEAR

SET DEFAULT TO (M.cCheminUtilisateur + M.cLogiciel)

RETURN M.cRacine

*---------------------------------------------------------------- CopyOnlyNew

PROCEDURE CopyOnlyNew

PARAMETERS cSource, cDestination, cExtension

PRIVATE I, N, P, aSource, aDest

DIMENSION aSource(1,5)
DIMENSION aDest(1,5)

IF EMPTY(M.cSource)
   M.cSource = SET("DIRECTORY")
ENDIF

IF EMPTY(M.cExtension)
   M.cExtension = "*.*"
ENDIF



N = ADIR(aDest, ADDBS(M.cDestination) + M.cExtension)

FOR I = 1 TO ADIR(aSource, ADDBS(M.cSource) + M.cExtension)
    IF N > 0
       P = ASCAN(aDest,aSource(I,1))
       IF P > 0
          P = ASUBSCRIPT(aDest,P,1)
          IF (aSource(I,3) = aDest(P,3)) AND (aSource(I,4) = aDest(P,4))
             LOOP && Même date et même heure
          ENDIF
       ENDIF
    ENDIF

    COPY FILE (ADDBS(M.cSource) + aSource(I,1)) ;
           TO (ADDBS(M.cDestination) + aSource(I,1))
ENDFOR

RETURN

*---------------------------------------------------------------- Type_Erreur

PROCEDURE Type_Erreur

LPARAMETERS nErr

DO CASE

   CASE M.nErr = 3     && Fichier en cours d'utilisation
        
   CASE M.nErr = 1705  && Accès au fichier refusé
        Alerte("Accès au fichier refusé" + CHR(13) + ;
               "Source : " + M.cSource+'\'+aSource(I,1) + CHR(13) + ;
               "Destination : " + (M.cDestination +'\' + aSource(I,1)))

   OTHERWISE
   
        Alerte(LTRIM(STR(LINENO())) + ' : ' + MESSAGE() + CHR(13) + ;
               MESSAGE(1) + CHR(13) + PROGRAM())
        CANCEL
ENDCASE

RETURN

*---------------------------------------------------------------- Alerte

PROCEDURE Alerte

LPARAMETERS cMessage

=MessageBox(IIF(VARTYPE(M.cMessage)='C',M.cMessage,TRANSFORM(M.cMessage)),48,"Attention")

RETURN


